







```

'*****
'*          Receiver          *
'*****

INCLUDE "modedefs.bas"

DEFINE          OSC          4          'Xtal frequency
DEFINE          PULSIN_MAX   968        '>968 RETURNS 0
DEFINE          DEBUG_REG    PORTB
DEFINE          DEBUG_BIT    7          'PORTB.7
DEFINE          DEBUG_MODE    1          'Inverted
DEFINE          DEBUG_BAUD   9600

RF              VAR          BYTE[4]
RXPin           VAR          PORTC.2
space           VAR          BYTE
i               VAR          BYTE
bits            VAR          BYTE
stx             VAR          WORD      'start of transmission

ADCON1 = 7

init:   RF[0]=0:RF[1]=0:RF[2]=0:RF[3]=0
        bits=0:i=0
        PULSIN RXPin, 1, stx
        IF (stx<792) THEN init
        WHILE RXPin=0:WEND          'wait pulse
        Loop:
            PULSIN RXPin, 0, space
            IF (space<40) OR (space>175) THEN init
            IF (space>75) THEN
                RF.0(i)=1          'set bit
            ENDIF
            i=i+1
        IF i<=31 THEN GOTO Loop
        IF RF[0]+RF[1]<>255 THEN init
        IF RF[2]+RF[3]<>255 THEN init
        FOR i = 0 TO 3
            DEBUG (RF[i] REV 8)
        NEXT
        GOTO init

END

```

```

'*****
'*           Transmitter           *
'*****

INCLUDE "modedefs.bas"

DEFINE          OSC                4                'Xtal frequency
RF               VAR BYTE [4]
Copies           VAR BYTE          'RF copies
c               VAR BYTE          'loop index (RF copies)
b               VAR BYTE          'loop index (RF[b])
i               VAR BYTE          'bit index
wb              VAR BYTE          'work byte
PulsPin         VAR                PORTC.2

ADCON1 = 7
Copies = 4

SendRF: RF[0]=80:RF[1]=~RF[0]:RF[2]=66:RF[3]=~RF[2]
  LOW PulsPin
  FOR c=1 TO Copies
    PULSOUT PulsPin, 880                '8.8mS lead-in pulse
    PAUSEUS 4400                      '4.4mS space
    FOR b=0 TO 3
      wb=RF[b]
      FOR i=0 TO 7                    'LSB first
        PULSOUT PulsPin, 50          '0.5mS pulse
        IF wb.0=1 THEN
          PAUSEUS 1500                '1.5mS space
        ELSE
          PAUSEUS 500                 '0.5mS space
        ENDIF
        wb=wb>>1
      NEXT
    NEXT
    PULSOUT PulsPin, 50                '0.5mS pulse
    PAUSE 40:                          '40mS GAP
  NEXT
  SLEEP 15                            '15 SEC DELAY
  GOTO SendRF

END

```